

## 9/4.3

# Biorythmes

Des études statistiques ont permis de mettre en évidence trois cycles de vie spécifiques à chaque individu. Il s'agit des cycles :

- physique (23 jours),
- émotionnel (28 jours),
- intellectuel (31 jours).

Ces cycles sinusoïdaux ont une origine commune le jour de la naissance de l'intéressé. Ils indiquent une période positive si, pour un jour donné, l'ordonnée est positive, une période négative si l'ordonnée est négative et une période critique si l'ordonnée est nulle (barricadez-vous chez vous car tout peut arriver ces jours-là !).

```

1000 MODE 1:PRINT"Biorythmes"
1010 '=====
1020 'Initialisation
1030 '=====
1040 DIM T(12)
1050 FOR I=1 TO 12
1060   READ T(I)
1070 NEXT I
1080 DATA 31,28,31,30,31,30,31,31,30,31,30,31
1090 '=====
1100 'Entree des donnees
1110 '=====
1120 INPUT"Jour de naissance ";JN
1130 INPUT"Mois de naissance ";MN
1140 INPUT"Annee de naissance ";AN
1150 PRINT:INPUT"Mois du trace ";MB
1160 INPUT"Annee du trace ";AB
1170 '=====
1180 'Trace des biorythmes
1190 '=====
1200 '
1210 '-----
1220 'Nombre de jours ecoules depuis la naissance
1230 '-----
1240 IF (AN/4)=INT(AN/4) THEN IN=1 ELSE IN=0
1250 IF IN=1 AND MN=2 THEN FIN=29 ELSE FIN=T(MN)
1260 FOR I=JN TO FIN
1270   TT=TT+1
1280 NEXT I
1290 FOR I=MN+1 TO 12
1300   IF IN=1 AND I=2 THEN TT=TT+29 ELSE TT=TT+T(I)
1310 NEXT I
1320 '

```

```

1330 FOR I=2 TO AB-AN
1340   B=AN+I-1
1350   IF B/4=INT(B/4) THEN TT=TT+1
1360   TT=TT+365
1370 NEXT I
1380 '
1390 IF AB/4=INT(AB/4) THEN BI=1 ELSE BI=0
1400 FOR I=1 TO MB-1
1410   IF BI=1 AND I=2 THEN TT=TT+29 ELSE TT=TT+T(I)
1420 NEXT I
1430 '-----
1440 'Affichage
1450 '-----
1460 CLS
1470 FOR I=1 TO T(MB)
1480   A=SIN(2*PI*(TT+I)/23)
1490   LOCATE I+3,11-INT(9*A):PRINT"*"
1500   C=SIN(2*PI*(TT+I)/33)
1510   LOCATE I+3,11-INT(9*C):PRINT"+"
1520   B=SIN(2*PI*(TT+I)/28)
1530   LOCATE I+3,11-INT(9*B):PRINT"."
1540 NEXT I
1550 FOR I=1 TO 7
1560   READ X,Y,D
1570   LOCATE X,Y:PRINT D
1580 NEXT I
1590 DATA 3,21,1,7,21,5,12,21,10,17,21,15,22,21,20,27,21,25,32,21,30
1600 '
1610 'Axes
1620 '
1630 LOCATE 4,11:PRINT"-----"
1640 FOR I=1 TO 20
1650   LOCATE 3,I+1:PRINT"@"
1660 NEXT I
1670 '=====
1680 'Legende
1690 '=====
1700 LOCATE 5,22:PRINT"* = Physique . = Affectif"
1710 LOCATE 5,23:PRINT"+ = Intellectuel"

```

Lignes 1000 à 1080 : Initialisation  
 Lignes 1090 à 1160 : Saisie des données pour tracer le biorythme  
 Lignes 1210 à 1420 : Calculs pour initialiser le biorythme courant  
 Lignes 1430 à 1710 : Affichage du biorythme

Exemple de biorythme issu du présent programme.

